```
> restart;
  with(PDEtools):
```

Stencils for the 2D Laplacian

The purpose of this worksheet is to introduce the five-point and nine-point stencils for the Laplacian in two dimensions. The Laplacian is defined as:

```
> laplacian := diff(u(x,y),x,x) + diff(u(x,y),y,y);
```

$$laplacian := \frac{\partial^2}{\partial x^2} u(x, y) + \frac{\partial^2}{\partial y^2} u(x, y) \tag{1}$$

We make use of the following procedure to generate centered stencils for each of the derivatives:

```
> centered_stencil := proc(r,N,{direction := x})
      local n, stencil, vars, beta_sol:
      n := floor(N/2):
      if (direction = y) then:
         stencil := D[2$r](u)(x,y) - add(beta[i]*u(x,y+i*h),i=-n..
  n);
         vars := [u(x,y),seq(D[2$i](u)(x,y),i=1..N-1)];
      else:
         stencil := D[1$r](u)(x,y) - add(beta[i]*u(x+i*h,y),i=-n..
  n);
         vars := [u(x,y),seq(D[1$i](u)(x,y),i=1..N-1)];
      fi:
      beta_sol := solve([coeffs(collect(convert(series(stencil,h,
  N),polynom),vars,'distributed'),vars)]):
      stencil := subs(beta_sol,stencil);
      convert(stencil = convert(series(stencil,h,N+2),polynom),
  diff);
  end proc:
```

Here are the particular 3-point stencils we will use:

```
> x_stencil := isolate(lhs(centered_stencil(2,3,direction=x)),diff
  (u(x,y),x,x));
  y_stencil := isolate(lhs(centered_stencil(2,3,direction=y)),diff
  (u(x,y),y,y));
```

$$x\_stencil := \frac{\partial^2}{\partial x^2} u(x, y) = \frac{u(x-h, y)}{h^2} - \frac{2\, u(x, y)}{h^2} + \frac{u(x+h, y)}{h^2}$$

$$y\_stencil := \frac{\partial^2}{\partial y^2} u(x, y) = \frac{u(x, y-h)}{h^2} - \frac{2\, u(x, y)}{h^2} + \frac{u(x, y+h)}{h^2} \tag{2}$$

Substituting these into the Laplacian yields the standard five-point stencil:

```
> fivepoint := subs(x_stencil,y_stencil,laplacian);
```

$$fivepoint := \frac{u(x-h, y)}{h^2} - \frac{4\, u(x, y)}{h^2} + \frac{u(x+h, y)}{h^2} + \frac{u(x, y-h)}{h^2} + \frac{u(x, y+h)}{h^2} \tag{3}$$

We can also define a nine-point stencil by adding contributions from the "corners": u(x+h,y+h), u(x+h,y-h), u(x-h,y+h) and u(x-h,y-h). This gives the general form of a nine-point stencil

```
> ninepoint := add(add(beta[i,j]*u(x+i*h,y+j*h),i=-1..1),j=-1..1);
```

$$ninepoint := \beta_{-1,-1}\, u(x-h, y-h) + \beta_{0,-1}\, u(x, y-h) + \beta_{1,-1}\, u(x+h, y-h) \tag{4}$$

$$+ \beta_{-1,0}\, u(x-h, y) + \beta_{0,0}\, u(x, y) + \beta_{1,0}\, u(x+h, y) + \beta_{-1,1}\, u(x-h, y+h)$$

$$+ \beta_{0,1}\, u(x, y+h) + \beta_{1,1}\, u(x+h, y+h)$$

Before, when we have defined such a stencil we have expanded in a Taylor series, subtracted off the quantity we wish to approximate, and demanded that all terms up to some error or order h^p vanish from which we determined the beta's. However, in this case demanding that the error be of order h^2 won't be enough to fix the beta's uniquely, and if we go up to h^3 there will be no solution. So, the standard ninepoint stencil is defined by the *choice:*

```
> ninepoint_assumptions := [beta[0,0] = -10/3/h^2, beta[0,1] =
  2/3/h^2, beta[0,-1] = 2/3/h^2, beta[1,0] = 2/3/h^2, beta[-1,0] =
  2/3/h^2, beta[1,1] = 1/6/h^2, beta[1,-1] = 1/6/h^2, beta[-1,1] =
  1/6/h^2, beta[-1,-1] = 1/6/h^2];
  ninepoint := subs(ninepoint_assumptions,ninepoint);
```

$$ninepoint\_assumptions := \left[\beta_{0,0} = -\frac{10}{3\,h^2}, \beta_{0,1} = \frac{2}{3\,h^2}, \beta_{0,-1} = \frac{2}{3\,h^2}, \beta_{1,0} = \frac{2}{3\,h^2}, \beta_{-1,0} = \frac{2}{3\,h^2}, \beta_{1,1} = \frac{1}{6\,h^2}, \beta_{1,-1} = \frac{1}{6\,h^2}, \beta_{-1,1} = \frac{1}{6\,h^2}, \beta_{-1,-1} = \frac{1}{6\,h^2}\right]$$

$$ninepoint := \frac{1}{6}\frac{u(x-h, y-h)}{h^2} + \frac{2}{3}\frac{u(x, y-h)}{h^2} + \frac{1}{6}\frac{u(x+h, y-h)}{h^2} + \frac{2}{3}\frac{u(x-h, y)}{h^2} - \frac{10}{3}\frac{u(x, y)}{h^2} + \frac{2}{3}\frac{u(x+h, y)}{h^2} + \frac{1}{6}\frac{u(x-h, y+h)}{h^2} + \frac{2}{3}\frac{u(x, y+h)}{h^2} + \frac{1}{6}\frac{u(x+h, y+h)}{h^2} \tag{5}$$

We can calculate the error in the two stencil in the standard way:

```
> fivepoint_error := convert(series(laplacian-fivepoint,h),diff);
  ninepoint_error := convert(series(laplacian-ninepoint,h),diff);
```

$$fivepoint\_error := \left(-\frac{1}{12}\frac{\partial^4}{\partial y^4}u(x, y) - \frac{1}{12}\frac{\partial^4}{\partial x^4}u(x, y)\right)h^2 + O(h^4)$$

$$ninepoint\_error := \left(-\frac{1}{12}\frac{\partial^4}{\partial x^4}u(x, y) - \frac{1}{6}\frac{\partial^4}{\partial y^2\,\partial x^2}u(x, y) - \frac{1}{12}\frac{\partial^4}{\partial y^4}u(x, y)\right)h^2 + O(h^4) \tag{6}$$

Notice that they are both errors of order h^2. But the error of the ninepoint stencil has a very interesting form. Let's calculate the Laplacian of the Laplacian of u:

```
> laplacianoflaplacian := Laplacian^2 = expand(subs(u(x,y)=
  laplacian,laplacian));
```

$$laplacianoflaplacian := Laplacian^2 = \frac{\partial^4}{\partial x^4}u(x, y) + 2\left(\frac{\partial^4}{\partial y^2\,\partial x^2}u(x, y)\right) + \frac{\partial^4}{\partial y^4}u(x, y) \tag{7}$$

But this is just proportional to the h^2 constribution to the ninepoint error; i.e.,

```
> ninepoint_error := subs(isolate(laplacianoflaplacian,diff(u(x,y),
  x,x,x,x)),ninepoint_error);
```

$$ninepoint\_error := -\frac{1}{12}\,Laplacian^2\,h^2 + O(h^4) \tag{8}$$

Hence, if the Laplacian of u is itself zero, then the h^2 part of the ninepoint error vanishes and we have a much more accurate stencil. This indeed occurs when we are solve Laplace's equation, which is

```
> laplace_eq := laplacian = 0;
  possion_eq := laplacian - f(x,y) = 0;
```

$$laplace\_eq := \frac{\partial^2}{\partial x^2} u(x, y) + \frac{\partial^2}{\partial y^2} u(x, y) = 0$$

$$possion\_eq := \frac{\partial^2}{\partial x^2} u(x, y) + \frac{\partial^2}{\partial y^2} u(x, y) - f(x, y) = 0 \tag{9}$$

We will also be interested in the solution to the Poisson equation, which is also given above. In this, f is supposed to be a known function. We can use either the fivepoint or ninepoint stencils in each of these equations:

```
> laplace_stencil_5 := fivepoint = 0;
  laplace_stencil_9 := ninepoint = 0;
  possion_stencil_5 := fivepoint - f(x,y) = 0;
  possion_stencil_9 := ninepoint - f(x,y) = 0;
```

$$laplace\_stencil\_5 := \frac{u(x-h, y)}{h^2} - \frac{4\, u(x, y)}{h^2} + \frac{u(x+h, y)}{h^2} + \frac{u(x, y-h)}{h^2} + \frac{u(x, y+h)}{h^2} = 0$$

$$laplace\_stencil\_9 := \frac{1}{6} \frac{u(x-h, y-h)}{h^2} + \frac{2}{3} \frac{u(x, y-h)}{h^2} + \frac{1}{6} \frac{u(x+h, y-h)}{h^2} + \frac{2}{3} \frac{u(x-h, y)}{h^2} - \frac{10}{3} \frac{u(x, y)}{h^2} + \frac{2}{3} \frac{u(x+h, y)}{h^2} + \frac{1}{6} \frac{u(x-h, y+h)}{h^2} + \frac{2}{3} \frac{u(x, y+h)}{h^2} + \frac{1}{6} \frac{u(x+h, y+h)}{h^2} = 0$$

$$possion\_stencil\_5 := \frac{u(x-h, y)}{h^2} - \frac{4\, u(x, y)}{h^2} + \frac{u(x+h, y)}{h^2} + \frac{u(x, y-h)}{h^2} + \frac{u(x, y+h)}{h^2} - f(x, y) = 0$$

$$possion\_stencil\_9 := \frac{1}{6} \frac{u(x-h, y-h)}{h^2} + \frac{2}{3} \frac{u(x, y-h)}{h^2} + \frac{1}{6} \frac{u(x+h, y-h)}{h^2} + \frac{2}{3} \frac{u(x-h, y)}{h^2} - \frac{10}{3} \frac{u(x, y)}{h^2} + \frac{2}{3} \frac{u(x+h, y)}{h^2} + \frac{1}{6} \frac{u(x-h, y+h)}{h^2} + \frac{2}{3} \frac{u(x, y+h)}{h^2} + \frac{1}{6} \frac{u(x+h, y+h)}{h^2} - f(x, y) = 0 \tag{10}$$

As usual, the error in a given stencil is calculated by expanding about h = 0 and substituting in the original PDE. We get

```
> err_laplace_stencil_5 := convert(dsubs(isolate(laplace_eq,diff(u
  (x,y),x,x)),simplify(convert(series(lhs(laplace_stencil_5),h=0,
```

```
4),diff))),D);
err_laplace_stencil_9 := convert(dsubs(isolate(laplace_eq,diff(u
(x,y),x,x)),simplify(convert(series(lhs(laplace_stencil_9),h=0,
10),diff))),D);
err_poisson_stencil_5 := convert(dsubs(isolate(possion_eq,diff(u
(x,y),x,x)),simplify(convert(series(lhs(possion_stencil_5),h=0,
4),diff))),D);
err_poisson_stencil_9 := convert(dsubs(isolate(possion_eq,diff(u
(x,y),x,x)),simplify(convert(series(lhs(possion_stencil_9),h=0,
6),diff))),D);
```

$$err\_laplace\_stencil\_5 := \frac{1}{6} D_{2,2,2,2}(u)(x,y)\, h^2 + O(h^4)$$

$$err\_laplace\_stencil\_9 := \frac{1}{3024} D_{2,2,2,2,2,2,2,2}(u)(x,y)\, h^6 + O(h^8)$$

$$err\_poisson\_stencil\_5 := \left(\frac{1}{6} D_{2,2,2,2}(u)(x,y) - \frac{1}{12} D_{2,2}(f)(x,y) + \frac{1}{12} D_{1,1}(f)(x,y)\right) h^2 + O(h^4)$$

$$err\_poisson\_stencil\_9 := \left(\frac{1}{12} D_{2,2}(f)(x,y) + \frac{1}{12} D_{1,1}(f)(x,y)\right) h^2 + \left(\frac{1}{360} D_{2,2,2,2}(f)(x,y) + \frac{1}{90} D_{1,1,2,2}(f)(x,y) + \frac{1}{360} D_{1,1,1,1}(f)(x,y)\right) h^4 + O(h^6) \quad \textbf{(11)}$$

As we expected, the error for the ninepoint stencil on the Laplace equation is much more accurate than for the fivepoint stencil. The error terms for the Poisson equation are interesting, They both appear to be of order h^2, but the lowest order error terms in the ninepoint version are expressed entirely in terms of the known function f. In other words, we can calculate them explicitly. This suggests we modify the original ninepoint stencil for the Poisson equation as follows:

```
> possion_stencil_9_modified := ninepoint - f(x,y) - convert
(err_poisson_stencil_9,polynom) = 0;
```

$$possion\_stencil\_9\_modified := \frac{1}{6}\frac{u(x-h,y-h)}{h^2} + \frac{2}{3}\frac{u(x,y-h)}{h^2} + \frac{1}{6}\frac{u(x+h,y-h)}{h^2} + \frac{2}{3}\frac{u(x-h,y)}{h^2} - \frac{10}{3}\frac{u(x,y)}{h^2} + \frac{2}{3}\frac{u(x+h,y)}{h^2} + \frac{1}{6}\frac{u(x-h,y+h)}{h^2} + \frac{2}{3}\frac{u(x,y+h)}{h^2} + \frac{1}{6}\frac{u(x+h,y+h)}{h^2} - f(x,y) - \left(\frac{1}{12} D_{2,2}(f)(x,y) + \frac{1}{12} D_{1,1}(f)(x,y)\right) h^2 - \left(\frac{1}{360} D_{2,2,2,2}(f)(x,y) + \frac{1}{90} D_{1,1,2,2}(f)(x,y) + \frac{1}{360} D_{1,1,1,1}(f)(x,y)\right) h^4 = 0 \quad \textbf{(12)}$$

The error in this modified Poisson stencil is now h^6, just like for the ninepoint Laplace equation stencil.

```
> err_poisson_stencil_9_modified := convert(dsubs(isolate
  (possion_eq,diff(u(x,y),x,x)),simplify(convert(series(lhs
  (possion_stencil_9_modified),h=0,10),diff))),D);
```

$$err\_poisson\_stencil\_9\_modified := \left(\frac{1}{3024} D_{2,2,2,2,2,2,2,2}(u)(x,y) - \frac{17}{60480} D_{2,2,2,2,2,2}(f)(x,y) + \frac{1}{1344} D_{1,1,2,2,2,2}(f)(x,y) + \frac{5}{12096} D_{1,1,1,1,2,2}(f)(x,y) + \frac{1}{20160} D_{1,1,1,1,1,1}(f)(x,y)\right) h^6 + \mathrm{O}\left(h^8\right) \quad \textbf{(13)}$$